

1 最初の一步

ここでは Maple の基礎的な操作法と本書の記法を述べます。


1.1 Maple の起動法

PC ではほかのソフトウェアと同様の立ち上げ方で起動します。user interface は Java base で、ほとんどの OS 環境で同じ面構えです。linux では xmaple とすると GUI 版の maple が立ち上がります。maple だけですと普通の terminal 上での character 版が立ち上がります。これを unix の redirection を使って filter として機能させることも可能です。

1.2 簡単な演算

それでは簡単な計算を実行させてみましょう。プロンプト (">") に続けていくつかのコマンドを打ち込んでみてください。

```
> 1+1; [enter]
2
> factor((x^2-3*x+2)); [shift+enter]
> 100!; [enter]
(x - 1)(x - 2)
9332621544394415268169923885626670049071596826438162
1468592963895217599993229915608941463976156518286253
69792082722375825118521091686400000000000000000000000
> plot(tanh(x), x=-5..5); [enter]
```



- enter と shift+enter は違った意味を持ちます。enter は入力、shift+enter は改行です。複数行にまたがる入力では shift+enter で改行を挿入します。
- 入力領域(デフォルトでは赤く表示されています)のどこかにカーソルを持っていきクリックして、enter をいれればその領域すべてを一度に入力したこ

とになります。

- 入力の順番は enter を入れた順番であり，画面の上下とは関係ありません．また入力領域のどの位置にカーソルがあっても enter でその領域全部を入力したことになります．
- 最後の ; (セミコロン) を忘れがちです．セミコロンはコマンドの区切りを表します．
- 出力させたくないときには最後の ; を : (コロン) にすれば，なにも出力しません．ただし，内部での代入は実行されています．
- これ以降の記述では最後の (enter) や (shift+enter) を省きます．
- これ以降の記述で続けてプロンプト (">") が表示されているのは，shift+enter で改行が入力されている，ひとかたまりの入力領域を意味します．

1.3 間違い修正


打ち間違いなどの訂正はアローキー，あるいはマウスのクリックによってできます．後は訂正して enter を入れれば入力されます．ある領域を選択して，削除・カットおよびペーストなどの作業もマウスを使ってできます．例えばサイン関数を $-\pi$ から $+\pi$ までプロットしようとした時

```
> plot(sin(x),x=-pi..pi);
```

```
Error, (in plot) range values must be real constants
```

という警告が出ました．これは Maple では大文字と小文字を区別しているためにおこったことです．そこで， $\pi \rightarrow \text{Pi}$ と修正すると無事表示されます．

```
> plot(sin(x),x=-Pi..Pi);
```



1.4 ヘルプファイル

> ?help;

でキーワードに関するヘルプが表示されます。その他, ??や?index あるいは? (キーワードの最初の一部) を使って, 類推によってキーワードの情報を取り出すことができます。その他の help に関する操作はメニューバーの『Help』にいくつか用意されています。

> ?plot

としてみてください。簡単な説明に続いて,

- Calling Sequence: (呼び出し)
- Parameters: (引数の説明)
- Description: (詳しい解説)
- Examples: (使用例)
- See Also: (関連する項目)

があります。記述は一部日本語に訳されています。英語が分からなくても Examples (使用例) を参考にすればだいたい予測できます。

2 簡単なコマンド

簡単な数値の代入と関数, 関数の定義と式の変形, グラフのプロットを扱います.

2.1 変数の代入とキャンセル

変数への数値の代入は

```
> mass:=10;
```

```
mass := 10
```

によって行われます. 式の定義も同様に行えます.

```
> force:=-mass*accel;
```

```
force := -10 accel
```

直前の結果を参照するには%を使います.

```
> exp1:=%;
```

```
exp1 := -10 accel
```

これら一度数値を入れた変数を元へ戻すには

```
> restart;
```

によって行います. これで起動初期のなにも入力されていない状態に戻ります. ひとつの定義だけを初期状態に戻したいときには' (シングルクォート) をもちいて

```
> mass:='mass';
```

```
mass := mass
```

によって行います. これによって,

```
> force:=-mass*accel;
```

```
force := -mass accel
```

となります. 一時的な代入は subs で行います.

```
> subs(mass=10,accel=14,force);
```

```
-140
```

こうすると,

```
> force;
```

```
-mass accel
```

とそれぞれの変数が数値を取るのではなく, 変数のままで扱われます. 逆に一時的に値ではなく変数そのものを使用するにもシングルクォートを使います.

```
> x:=2;y:=3;
```

```
x := 2
```

```
y := 3
```

```
> f:='x+y'; g:=x+y;
```

$$f := x + y$$
$$g := 5$$

(注：続けて入力される方はこちら辺でrestart をかけてください。以下では数値を代入した変数を，free の変数として使っています。そのまま入力続けると叱られます)

2.2 関数

よく使う関数はそのままの形で使えます。三角関数 (trigonometric functions) はラジアンで入れてください。log (もちろん ln も) は自然対数です。底をあらわにするときは

```
> log[2](5);
```

$$\frac{\ln(5)}{\ln(2)}$$

としてください。数値として取り出したいときには

```
> evalf(%);(evaluating float の略)
```

2.321928094

Maple が提供する膨大な数の関数から，目的とするものを探しだすには help を使って下さい。以下に関数等の index を表示する keywords をまとめておきます。

? inifens - 起動時から認識されている関数

? index[package] - 関連する関数を集めたパッケージです。微分方程式 (DETools), 線形代数 (linalg), プロット関係の関数 (plots) 等があります。

? index[function] - Maple の標準関数。

これらの関数は起動時から読み込まれている関数と，ユーザーが呼び出さなければならない関数とがあります。呼び出しが必要な時には

```
> with(plots);
```

などでおこなってください。日本語から引くためのホームページが

<http://ist.ksc.kwansei.ac.jp/~nishitani/Maple/JtoE.html>

にあります。まだ構築中ですが...

2.3 ユーザー関数の定義

単純なユーザー関数の定義は次の2種類を使います。

i) 矢印による定義.

ii) unapply による定義.

矢印による定義は普通の入力のようにして,

```
> f1:=x->-x*ln(exp(-1/x)/(1-exp(-1/x)));
```

$$f1 := x \mapsto -x \ln \left(e^{-x^{-1}} \left(1 - e^{-x^{-1}} \right)^{-1} \right)$$

unapply は一度求めた式を関数として定義するときに使います. たとえば以下では eq1 という式に入っている定義にしたがって, x を変数とする f1 という関数と定義しています.

```
> eq2:=(1+exp(-1/T))/(1-exp(-1/T));
```

```
f2:=unapply(eq2,T);
```

$$eq2 := \left(1 + e^{-T^{-1}} \right) \left(1 - e^{-T^{-1}} \right)^{-1}$$

$$f2 := T \mapsto \left(1 + e^{-T^{-1}} \right) \left(1 - e^{-T^{-1}} \right)^{-1}$$

まちがって数式に対して矢印での定義をすると

```
> f3:=T->eq2;
```

```
f3(3);
```


$$f3 := T \mapsto eq2 \\ \left(1 + e^{-T^{-1}} \right) \left(1 - e^{-T^{-1}} \right)^{-1}$$

となり変数を変数とみなしてくれません.

2.4 プロット


複雑な関数も Maple だとすぐに視覚化してくれます. 先程のユーザー定義した関数を使って, 関数が実際にどのような形をしているか plot させてみましょう.

```
> plot(f1(x),x=0..10);
```




となります. 2つの関数を一つの領域にプロットするには

```
> plot(f1(x),f2(x),x=0..1);
```



2変数の場合には plot3d を使います.

```
> plot3d(sin(x)*cos(y), x=-Pi..Pi, y=-Pi..Pi);
```



オプションや特殊な plotting 法があります. plot に対する一部の簡単な操作 (視点の変更, 表面の加工, 軸の挿入) はメニューバーからできます. さらに

```
> with(plots):
```

で呼び出される plots package や plottools package には多くの便利な表示関数が用意されています.

2.5 方程式の解

solve で方程式の解が求まります.

```
> eqset:={x+y=1,y=1+x^2};
```

$$eqset := \{x + y = 1, y = 1 + x^2\}$$

```
> solve(eqset, {x,y});
```

$$\{x = 0, y = 1\}, \{x = -1, y = 2\}$$

これだけでは

```
> x;y;
```

のように変数 x, y へ代入されていません。これを代入するには `assign` を使い

```
> solset:=solve(eqset,{x,y});
      solset := {x = 0, y = 1}, {x = -1, y = 2}
> solset[1];
      {x = 0, y = 1}
> assign(solset[1]);
> x;y;
```

0

1

のようにします。代数的に解けない方程式に対しても数値的に解く関数 `fsolve(f(x)=0,x)` があります。

2.6 微分

微分は `diff` によって行います。

```
> diff(x^2,x);
      2x
> diff(y^2*x^2,x,x);
      2y^2
```

関数の一般形のままでも形式的な微分を表示してくれます。

```
> c:=(x,t)->X(x)*T(t);
      c := (x, t) -> X(x) T(t)
> diff(c(x,t),t);
> diff(c(x,t),x,x);
      X(x) (∂/∂t T(t))
      (∂²/∂x² X(x)) T(t)
```

2.7 積分

不定積分、定積分はそれぞれ

```
> int(ln(x),x);
      x ln(x) - x
> int(sin(x),x=-Pi..0);
```

-2

などで求めます。int を integrate としても同じ結果を得ます。積分公式がないと求められないような関数も

```
> eq:=x^2/sqrt(1-x^2);  
> int(eq,x);
```

$$eq := \frac{x^2}{\sqrt{1-x^2}} - \frac{1}{2}x\sqrt{1-x^2} + \frac{1}{2}\arcsin(x)$$

```
> eq2:=exp(-x^2);  
> int(eq2,x=-z..z);
```

$$eq2 := e^{(-x^2)} \sqrt{\pi} \operatorname{erf}(z)$$

という具合です。積分を実行するのではなく、積分記号だけを表示させたいときには Int とします。

演習問題

1. 二つの関数 $x, \cos(x)$ を $x=-5..5$ でプロットせよ。
2. 新しい関数 $f(x)=x-\cos(x)$ を定義して、 $x=-5..5$ で表示せよ。
3. fsolve を用いて、 $x-\cos(x)=0$ を解け。
4. アインシュタイン結晶の内部エネルギーは

$$E(T) = \frac{1 + \exp(-1/T)}{1 - \exp(-1/T)} \quad (1)$$

で表される (T は規格化した温度)。その温度依存性をプロットせよ。また、比熱 ($C(T)=dE/dT$) を求め、その温度依存性をプロットせよ。